
C[ Language Concrete Syntax

1 C[ Syntax

In the specification below, orange characters are terminals. Integers, identifiers, and comments are specified
using regular expressions (extended with some common notation) while the rest of the language is specified
using a context-free grammar.

n ∈ Integer ::= -?[0–9]+

id ∈ Identifier ::= [a–zA–Z]+[0–9a–zA–Z]?

comment ∈ Comment ::= //.?\n

An integer is a non-empty sequence of digits, optionally preceded by a negative sign. An identifier is a
non-empty sequence of alphanumeric characters that must begin with an alphabetic character. A comment
starts with ‘//’ and ends at the next newline.

aexp ∈ ArithmeticExp ::= n | id | aexp + aexp | aexp – aexp | aexp * aexp | (aexp)

rexp ∈ RelationalExp ::= aexp < aexp | aexp = aexp | aexp <= aexp | rexp && rexp

| rexp || rexp | !rexp | [rexp]

An arithmetic expression is an integer, variable, or two arithmetic expressions combined with a plus, minus,
or times operator. A relational expression is two arithmetic expressions combined with a less-than, equals, or
less-than-or-equal-to operator; or two relational expressions combined with logical and or or; or a relational
expression preceded by not. For arithmetic expressions, the user can indicate grouping using parentheses.
For relational expressions, the grouping is indicated with brackets–distinguishing the use of brackets helps
keep parsing simple, as we are going to see later in class.

stmt ∈ Statement ::= assign | whileLoop | forLoop | cond

assign ∈ Assignment ::= id := aexp; | id := call ;

whileLoop ∈ WhileLoop ::= while rexp { block }
forLoop ∈ ForLoop ::= for var from aexp to aexp { block }

cond ∈ Conditional ::= if rexp { block } | if rexp { block } else { block }

A statement is an assignment, a while loop, a for loop, or a conditional. The left-hand side of an assignment
can be either an arithmetic expression or a function call. A conditional may or may not have an else branch.
The for loops are range-based rather than the C-style for loops.

type ∈ Type ::= int

decl ∈ Declaration ::= type id ;

decls ∈ Declarations ::= ε | decl decls

stmts ∈ Statements ::= ε | stmt stmts

block ∈ Block ::= decls stmts

A variable declaration is a type followed by a variable name. A block of statements is a (possibly empty)
sequence of declarations followed by a (possibly empty) sequence of statements.

call ∈ FunctionCall ::= id(args) | id()

args ∈ Arguments ::= aexp | aexp, args



A function call is the name of the function followed by a (possibly empty) sequence of arithmetic expression
arguments in parentheses separated by commas.

fundef ∈ FunctionDef ::= def id(optparams) : type { block return aexp; }
params ∈ Parameters ::= type id | type id , params

optparams ∈ OptionalParameters ::= ε | params

A function definition provides the function name; a (possibly empty) sequence of parameters giving the
type and name of each parameter, seperated by commas; the function’s return type; and the function body
consisting of a block of statements followed by a return statement.

fundefs ∈ FunctionDefs ::= ε | fundef fundefs

program ∈ Program ::= fundefs block output aexp;

A program consists of a (possibly empty) sequence of function definitions followed by a block of statements
followed by an output that will be printed by the program.

2 Example Program

This progam doesn’t do anything sensible, it’s just an example of the language syntax in use.

def foo(int a, int b) : int {

int x;

int y;

x := a + b * (a - b);

while (y <= x + 3) {

y := y + 10;

}

for (c from a to b * 2) {

y := y + x;

}

return y * 2;

}

def bar() : int { return 42; }

int t1;

int t2;

t1 := bar();

t2 := foo(t1 + 2, t1 * 2);

if (t1 < t2) { t1 := t2; }

output t1 + 42;


